

▼ 2. Egyváltozós nemlineáris egyenletek megoldása

Ebben a fejezetben az egyenletek gyökeinek közelítő meghatározására alkalmas módszerek közül megismerkedünk a legáltalánosabban használt Newton módszer (Newton-Rapson módszer) elméleti hátterével, alkalmazásának feltételeivel. A közelítő gyököket először lépésenként adjuk meg, majd egyre bonyolultabb eljárásokkal automatizáljuk a gyökkeresést, végül megismerkedünk a Maple beépített Newton eljárásával.

Olyan esetekben keressük az **egyenlet gyökeit**, illetve a **függvény zérushelyeit**, amikor megoldóképlet alkalmazása nem lehetséges.

A **gyök (megoldás)** meghatározása itt azt jelenti, hogy véges sok - és általában nem túl nagy számú - lépés után az elméletileg pontos értéktől adott hibakorlátnál kisebb mértékben tér el a gyököt közelítő érték.

Célunk olyan egyismeretlenes egyenlet gyökeinek meghatározása, amikor a megoldóképlet alkalmazása fel sem merülhet, illetve az egyenlet megoldása nem állítható elő zárt alakban.

▼ 2.1. Egyismeretlenes egyenletek megoldhatósága

Használt Maple parancsok: [solve](#), [fsolve](#), [RealDomain](#)

Vizsgáljuk meg a következő egyenleteket:

A megoldandó egyenlet	Az egyenlet típusa	Egzakt megoldás	A megoldás Maple-lel
1) $\sin(2x) - \cos(x) = 0$	Nem algebrai	$2 \sin(x) \cos(x) - \cos(x) = 0$ $\cos(x) (2 \sin(x) - 1) = 0$ $\cos(x)=0 \quad x = \frac{\pi}{2} + k\pi, k \in \mathbb{Z}$ $\sin(x) = \frac{1}{2} \quad x = \frac{\pi}{6} + 2k\pi, x = \frac{5\pi}{6} + 2k\pi, k \in \mathbb{Z}$	$egyenlet1 := \sin(2x) - \cos(x) = 0;$ $\sin(2x) - \cos(x) = 0$ $solve(egyenlet1, x)$ $\frac{1}{2}\pi, -\frac{1}{2}\pi, \frac{1}{6}\pi, \frac{5}{6}\pi$
2) $(x^2 + 4)(x^2 - 1) = 0$	Algebrai	$x^2 + 4 \neq 0$, így $x^2 - 1 = 0, x = \pm 1$	$egyenlet2 := (x^2 + 4)(x^2 - 1) = 0;$ $(x^2 + 4)(x^2 - 1) = 0$ $solve(egyenlet2, x);$ $-1, 1$ $with(RealDomain):$ $solve(egyenlet2, x);$ $-1, 1$
3) $x^3 - 9x^2 + 2x + 6$	Algebrai	Van megoldóképlet	$egyenlet3 := x^3 - 9x^2 + 2x + 6 = 0;$ $x^3 - 9x^2 + 2x + 6 = 0$

$= 0$			$\text{solve}(\text{egyenlet3}, x);$ $4 - \sqrt{22}, 4 + \sqrt{22}, 1$
4) $x^5 - 9x^2 = 6$	Algebrai	Nincs megoldóképlet, nincs egzakt megoldás	$\text{egyenlet4} := x^5 - 9x^2 + 6 = 0;$ $x^5 - 9x^2 + 6 = 0$ $\text{solve}(\text{egyenlet4}, x)$ $\text{RootOf}(_Z^5 - 9_Z^2 + 6, -0.7946447004),$ $\text{RootOf}(_Z^5 - 9_Z^2 + 6, 1.950736723),$ $\text{RootOf}(_Z^5 - 9_Z^2 + 6, 0.8453619345)$ $\text{fsolve}(\text{egyenlet4}, x);$ $-0.7946447002,$ $0.8453619348,$ 1.950736723 $\text{fsolve}(\text{egyenlet4}, x = 0..1);$ 0.8453619348
5) $e^x = 2 - x^2$	Nem algebrai	Nincs megoldóképlet, nincs egzakt megoldás	$\text{egyenlet5} := e^x = \sqrt{x + 1}$ $e^x = \sqrt{x + 1}$ $\text{solve}(\text{egyenlet5}, x)$ $0, -\frac{1}{2} \text{LambertW}(-2e^{-2}) - 1$ $\text{fsolve}(\text{egyenlet5}, x)$ $0.$ $\text{evalf}(\text{solve}(\text{egyenlet5}, x), 4)$ $0., -0.7968$

Megjegyzések:

- Az 1) és 2) egyenlet középiskolai feladat, analitikusan megoldható. A Maple a trigonometrikus egyenlet egy periódusba eső megoldásait adja meg. Alapesetben az alaphalmaz a komplex számok halmaza, a *RealDomain* csomag elindításával szűkíthetjük a megoldásokat a valós számok körére.
- A 3) egyenlet harmadfokú, megoldására van megoldóképlet, de ismerete túlmutat a középiskolai anyagon, a Maple megadja az egzakt megoldást.
- A 4) egyenlet ötödfokú, megoldására nincs megoldóképlet, így a *solve* parancs nem ad megoldást. (Galois francia matematikus, 1811-1832 megmutatta, hogy ötödfokú, vagy magasabb fokú polinomok zérushelyeinek meghatározására nincs megoldóképlet, tehát olyan eljárás amely a négy alapművelet és a gyökvonás segítségével szimbolikusan megadható, pontos értékeket ad. Az *fsolve* parancs numerikusan megadja az összes, illetve egy előre megadott intervallumba eső

megoldásokat.

- Az 5) egyenlet algebrai és nem-algebrai kifejezéseket is tartalmaz. Megoldásként egy pontos értéket, illetve egy üzenetet kapunk a második gyök létezéséről. Az *evalf* paraccsal ennek adott jegyre való közelítését kapjuk.

▼ 2.2. A Newton-módszer (Newton-iteráció) alkalmazásának feltételei, az algoritmus lépései

▼ 2.2.1. Definíció

Iteráció (sorozatos megközelítés): kiindulva valamely, a megoldáshoz "viszonylag közeli" elemből, lépésenként kiszámítjuk egy, a megoldáshoz konvergáló végtelen sorozat egymás utáni tagjait. $\square$

Az egy pontra támaszkodó iteráció általános alakja

$$x_{i+1} = F(x_i),$$


ahol F az iterációs függvény, $i=0,1, \dots$

Ha az iterációs eljárás konvergens, akkor a keresett ξ gyökre igaz, hogy


$$\xi = F(\xi).$$

A **Newton-módszer** lényege az, hogy az egyenlet megoldása helyett **egy függvény zérushelyét keressük**, úgy hogy a függvény zérushelyét érintőinek zérushelyével közelítjük.

Lépés 1. A $h(x) = g(x)$ egyenlet 2 oldalát egy-egy függvénynek $h(x)$, $g(x)$ tekintve, ábrázoljuk a függvényeket. A függvénygörbék metszéspontjainak abszcisszái megegyeznek az egyenlet (valós) gyökeivel.




Lépés 2. Írjuk át az egyenletet $f(x) = h(x) - g(x) = 0$ alakra, így az előzőekben meghatározott metszéspontok abszcisszái az $f(x)$ függvény zérushelyeinek felel meg.




Lépés 3. Határozzuk meg a következő feltételeknek megfelelő $[a, b]$ intervallumot:

- az intervallumban csak egy gyök van és $f(a) \cdot f(b) < 0$,
- $f(x)$ az $[a, b]$ intervallumban kétszer deriválható (tehát folytonos is),
- $f'(x) \neq 0$ ha $x \in [a, b]$ (a függvény monoton),
- $f''(x) \neq 0$ ha $x \in [a, b]$ (a függvénynek nincs inflexiós pontja).



Lépés 4. Húzzunk érintőt az $f(x)$ függvényhez az intervallum azon végpontjából, ahol a függvényérték előjele megegyezik a második derivált előjével (Mivel ez a 0. közelítés, jelöljük x_0 -al.) A gyököt az érintő x tengellyel való metszéspontjával közelítjük.



Lépés 5. Írjuk fel az $(x_0, f(x_0))$ pontban húzott érintőegyenes egyenletét:

$$y - f(x_0) = f'(x_0)(x - x_0)$$

Lépés 6. Határozzuk meg az érintőegyenes x tengellyel való metszéspontját, ez lesz az első közelítő gyök

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}$$

Lépés 7. Az eljárást az $(x_1, f(x_1))$ pontból kiindulva folytatjuk, mindaddig, amíg az egymást követő két gyök közelítés eltérése a megadott pontosságnál nem kisebb.

Belátható, hogy ha $x_0 = b$, akkor a $b, x_1, x_2, \dots, x_n, \dots$ sorozat monoton csökkenő módon az $f(x)$ függvény egyetlen $[a, b]$ -beli ξ zérushelyéhez tart. Ha az $x_0 = a$ végpontból indulunk ki, akkor az $a, x_1, x_2, \dots, x_n, \dots$ sorozatot kapjuk, amely monoton növekvőleg tart ξ -hez.

Az eljárást többféleképpen is leállíthatjuk, vagyis az eljárásból való kilépés feltétele, hogy az alábbi **kilépési feltételek** valamelyike nem teljesül:

1. Az egymást követő érték különbségére $|x_{n+1} - x_n| < \epsilon$ ahol $0 < \epsilon$ a hibakorlát.
2. $|f(x_n)| < \epsilon$.

Összefoglalva: A Newton-módszer algoritmus:

	Megvalósítás Maple-ben megadott lépésszámig	Megvalósítás Maple -ben megadott hibahatárig
Input : $x_0, \epsilon (\epsilon > 0)$. $n=1,$ $x_n = x_{n-1} - \frac{f(x_{n-1})}{f'(x_{n-1})}$ $n = n + 1$ amíg kilépési feltétel = hamis Output : x_n	x_0, k for i to k do $x[i] := (x[(i-1)] - f(x[(i-1)]) / D(f)(x[(i-1)]))$; end do ; $x_1, x_2, \dots, x_k$	$x_0, \epsilon, k, hiba$ for i to k while $hiba > \epsilon$ do $x[i] := (x[(i-1)] - f(x[(i-1)]) / D(f)(x[(i-1)]))$; $hiba := (x[i] - x[(i-1)])$; end do ; $x_1, x_2, \dots, x_n$


▼ 2.3. A Newton iteráció megvalósítása Maple-ben

▼ 2.3.1. Iteráció lépésenként

1. Példa: Oldjuk meg az $x^3 = 4 \cdot x^2 - 6$ egyenletet!


Lépés 1.

> $h := x \rightarrow x^3; g := x \rightarrow 4x^2 - 6; \text{plot}(\{h(x), g(x)\}, x = -1.4..2, y = -6..6)$



Az ábrából látszik, hogy a feladatnak 2 megoldása van a $[-2, 2]$ intervallumban. Írjuk át az egyenletet $f(x)=0$ formába, majd ezt a függvényt ábrázoljuk.

> $f := x \rightarrow x^3 - 4x^2 + 6; \text{plot}(f(x), x = -1.4..2)$



Lépés 3. Az ábráról megállapítható, hogy az egyenletnek egyik gyöke ξ_1 a $[-1.4, -1]$, egy másik gyöke ξ_2 az $[1.5, 2]$ intervallumba esik. Keressük ξ_1 értékét!


Vizsgáljuk a feltételek teljesülését! (Mivel a derivált adott helyen vett értékére vagyunk kíváncsiak, célszerű $D(f)(x)$ utasítással meghatározni a derivált értékét.)

> $első_derivált := D(f)(x)$; $masodik_derivált := D^{(2)}(f)(x)$

$$első_derivált := 3x^2 - 8x$$

$$masodik_derivált := 6x - 8$$

> $plot([f(x), első_derivált, masodik_derivált], x = -1.4 \dots -1, color = [red, blue, green], legend = [f(x), első_derivált, masodik_derivált])$



Az ábra jól mutatja, hogy az első derivált mindenütt pozitív, a második derivált mindenütt negatív az adott intervallumon.

Lépés 4. 5. 6. Mivel a második derivált negatív, ezért a negatív függvényértékű pontból kell kiindulni.

> $x0 := -1.5$; $x1 := x0 - f(x0) / D(f)(x0)$;

$$x0 := -1.5$$

$$x1 := -1.160000000$$

Lépés 7. Véges számú lépés végrehajtásához elég egyszerű *for* ciklust írunk. Meg kell adni a lépésszámot, és a kezdőértéket.

> $k := 8$; $x0 := -1.5$; $x_0 := x0$;

> **for** i **to** k **do** $x_i := evalf\left(x_{i-1} - \frac{f(x_{i-1})}{D(f)(x_{i-1})}\right)$ **end do** :

> $kozelitogyok := x[k]$

$$kozelitogyok := -1.086130198$$

Ha az előre megadott ϵ hibahatárig akarjuk elvégezni a számításokat a *while* ciklust használjuk.

> $k := 8$; $x0 := -1.5$; $x[0] := x0$; $\epsilon := 0.001$; $hiba := 100$:

> **for** i **to** k **while** $\epsilon < hiba$ **do**

$$x[i] := \text{evalf}\left(x[i-1] - \frac{f(x[i-1])}{D(f)(x[i-1])}\right) : \text{hiba} := \text{evalf}(x[i] - x[i-1])$$

end do :

> *hiba* := *hiba*; *kozelitogyok* := *x*[*k*]

hiba := 0.000005451

kozelitogyok := -1.086130198

▼ 2.3.2. Eljárások a Newton iterációra

Az **iteral** eljárás az iteráció egy lépését elvégző függvényt készít.

> *restart* : *with*(*plots*) :

> *iteral* := **proc**(*expr* : *algebraic*, *x* : *name*)
 local *iteration*;
 iteration := *x* - *expr* / *diff*(*expr*, *x*);
 unapply(*iteration*, *x*)
end proc :


2. Példa: Határozzuk meg az $e^x = 2 - x^2$ egyenlet pozitív gyökét!

> $f := x \rightarrow e^x + x^2 - 2 :$

> *Newton* := *iteral*(*f*(*x*), *x*)

$$\text{Newton} := x \rightarrow x - \frac{e^x + x^2 - 2}{e^x + 2x}$$

> *plot*([*f*(*x*), *D*(*f*)(*x*), *D*⁽²⁾(*f*)(*x*)], *x* = -2 .. 1, *color* = [*blue*, *red*, *green*], *thickness* = 2)



Az ábrából látható, hogy a pozitív gyök a $[0,1]$ intervallumba esik, ahol az első és a második derivált is pozitív. Kezdőértéknek tehát a zérushelynél nagyobb értéket kell választanunk, mert ekkor egyezik meg a második derivált előjele a függvényérték előjelével. Legyen a kezdőérték 1.0.

Adjuk meg a kezdőértéket (legyen ez x_0) és a lépések számát (k) !

> *x0* := 1.0; *x*[0] := *x0* : *k* := 6 :

> **for** *i* **to** *k* **do** *x*[*i*] := *Newton*(*x*[*i* - 1]) **end do**

x0 := 1.0

*x*₁ := 0.6358246729

$x_2 := 0.5431566927$

$x_3 := 0.5372973587$

$x_4 := 0.5372744493$

$x_5 := 0.5372744493$

$x_6 := 0.5372744493$

Készítsünk animációs ábrát is az eljárás "lefutásáról"!

> $x0 := 1.0$: $rajz := plot(f(x), x = x0 - 0.7 .. x0 + 0.1)$: $kep[0] := display(rajz)$:

> **for** i **to** k **do**

$x1 := Newton(x0)$;


$P := plot([[x0, 0], [x0, f(x0)], [x1, 0]], color = blue)$;

$kep[i] := plots[display]([kep[(i - 1)], P])$;

$x0 := x1$

end do:

> $display([kep[k]])$



A következő **Newtoniteral** eljárás addig fut, amíg vagy kisebb lesz két egymást követő érték különbsége az előírt mértéknél, vagy túl nem lépjük a megengedett lépésszámot. Az eljárás második része az animációhoz szükséges képsorozatot állítja elő.

> $restart : with(plots)$:

> **Newtoniteral** := **proc** ($f, x0, n, delta$)

local $x, graf1, graf2, p, p1, kep, z, y, mini, maxi, a, b$;

global i ;

$x := x0$;

$i := 1$;

$mini := 10^{10}$;

$maxi := -10^{10}$;

$z := x$;

$y := x - f(x) / D(f)(x)$;

while $i \leq n$ **and** $evalf(delta) \leq evalf(abs(y - z))$ **do**

$y := evalf(x - f(x) / D(f)(x))$;


```

    z := x;
    mini := min( evalf(x), evalf(mini) );
    maxi := max( evalf(x), evalf(maxi) );
    x := y;
    i := i + 1
end do;
a := min( mini, evalf(x0) ) - Float(2, -1);
b := max( evalf(x0), maxi ) + Float(2, -1);
graf1 := plot( f(t), t=a..b, color=blue);
kep||0 := display( graf1 );
i := 1;
x := x0;
y := x - f(x) / D( f ) ( x );
z := x;
while i ≤ n and evalf( delta ) ≤ evalf( abs(y - z) ) do
    y := x - f(x) / D( f ) ( x );
    z := x;
    p := plot( [ [x, 0], [x, f(x)], [y, 0]], thickness=1 );
    kep||i := display( [kep|| (i - 1), p] );
    x := y;
    i := i + 1;
    print( 'x'[i - 1] = evalf(x) )
end do
end proc:

```

3. Példa: Határozzuk meg az $f(x) = x^3 - 0.3x^2 + 1$ függvény zérushelyeit!

> $f := x \rightarrow x^3 - 0.3x^2 + 1$;
 $\text{plot}([f(x), D(f)(x), D^{(2)}(f)(x)], x = -1.1..1, y = -0.8..1.7, \text{color} = [\text{blue}, \text{red}, \text{green}])$




Az ábrából látszik, hogy a függvénynek (kék) egy zérushelye van, hiszen a derivált (piros) másodfokú, és egy "kis" szakasztól eltekintve pozitív, tehát a függvénynek a grafikonon nem látható része szigorúan monoton növekvő. Az is látszik, hogy a gyök a $[-1, -0.5]$ intervallumban van. A kezdőérték lehet például a -1, vagy ennél kisebb szám, mert ekkor a második derivált (zöld) megegyezik a függvény előjelével, és mind az első, mind a második

derivált előjele állandó a zérushelyet tartalmazó intervallumban.

Legyen a kezdőérték -2, a megengedett maximális lépésszám 10, a kívánt pontosság 10^{-8} .

```
> Digits := 12 : Newtoniteral(f, -2, 10, 10-8)  
       $x_1 = -1.37878787879$   
       $x_2 = -1.04321025530$   
       $x_3 = -0.924519905634$   
       $x_4 = -0.909565212175$   
       $x_5 = -0.909339283947$   
       $x_6 = -0.909339232865$   
       $x_7 = -0.909339232865$ 
```


```
> display([kep||(1..i-1)], insequence=false)
```



▼ 2.4. Példák a feltételvizsgálat fontosságára


4. Példa: Határozzuk meg az $f(x) = x - 2x^{\frac{1}{2}} + \sin(x)$ függvény legnagyobb zérushelyét!

```
> f := x → x - 2√x + sin(x); plot(f(x), x=0..15)  
       $f := x \rightarrow x - 2\sqrt{x} + \sin(x)$ 
```




Az ábra alapján arra következtethetünk, hogy a legnagyobb zérushely az $[5, 6]$ intervallumban van.

> `plot([f(x), D(f)(x), D(2)(f)(x)], x=5..6, color=[blue, red, green])`



> `Digits := 12 : Newtoniteral(f, 6, 10, 10-6)`
 $x_1 = 5.47058871628$
 $x_2 = 5.41769027340$
 $x_3 = 5.41679785537$
 $x_4 = 5.41679759339$

> `display([kep||(1..i-1)], insequence=false)`



A kezdőpont helytelen megválasztásának beláthatatlan következményei lehetnek! Legyen a kezdőérték 9.55.

> `Newtoniteral(f, 9.55, 10, 10-8)`

$$x_1 = 19.8251008611$$

$$x_2 = 11.0324933699$$

$$x_3 = 6.42535279392$$

$$x_4 = 5.48679240027$$


$$x_5 = 5.41828005653$$

$$x_6 = 5.41679831539$$

$$x_7 = 5.41679759340$$

$$x_8 = 5.41679759339$$

> `display([kep||(1..i-1)], insequence=false)`




A "közelítő" értékek kaotikusan viselkednek, bár egy idő után beáll a szigorú monoton csökkenés. Ha a kezdő érték 9.6

> `Newtoniteral(f, 9.6, 10, 10-8)`

Error, (in Newtoniteral) complex argument to max/min

A számolás során olyan értékek adódnak, ahol a függvény nincs is értelmezve. Ha csak a kapott

érintőket szemléltetjük, a következő ábrát kapjuk:



▼ 2.5. Gyakorlati alkalmazások


Használt Maple parancsok: [subs](#), [animate](#), [unapply](#)

5. Példa Kepler-probléma: A Nap körül keringő bolygók mozgását leíró egyenlet megoldása. A megoldásnak az elméleti alkalmazások (asztrofizika, atomfizika) mellett az időszámításban, a navigációban is fontos szerepe van.

A Kepler-törvények kimondják, hogy a bolygók – így a Föld is – Nap fókuszpontú ellipszis pályán mozognak, a mozgás során a bolygótól a Naphoz húzott vezérsugár egyenlő idők alatt egyenlő területeket sűrol, tehát a bolygó nem egyenletes mozgást végez.

A feladat ezen mozgás hely-idő összefüggésének megadása.

Kepler a problémára a következő megoldást találta. Tekintsünk egy olyan körpályát, melynek középpontja a földpálya középpontja, sugara pedig a földpálya fél nagytengelye. Ezen a pályán mozogjon egy test egyenletes sebességgel.



A valódi Föld $r=NF$ vezérsugarának meghosszabbításával érzékelhetővé válik a két objektum

mozgásának eltérése. A Föld a napközeltől indulva siet, majd lassulva ismét „találkozik” az egyenes mozgású égitesttel. A legnagyobb naptávolságot elhagyva a viszonyok megfordulnak: a Föld kezdetben lemarad, majd felgyorsulva utoléri az „etalont”. <http://hu.wikipedia.org/wiki/Kepler-probl%C3%A9ma>

Az ábrán a P pont a pályának az a pontja, ahol a bolygó a központi égitesthez legközelebb van: perihélium, vagyis $a=OP$ az ellipszis fél nagytengelye. A bolygó pillanatnyi helyzetét egyértelműen megadja az az ω szög, melyet az r vezérsugár az NP tengellyel bezár, ezt **valódi anomáliának** nevezzük.

A cél az $\omega = \omega(t)$ összefüggés meghatározása.

A feladat megoldásához vezető, úgynevezett Kepler-egyenlet:


>
$$\mu = \eta - e \cdot \sin(\eta) :$$

ahol $e = \frac{ON}{OP}$ - az úgynevezett **numerikus excentricitás**, ami a Föld esetén $e=0.01674$, μ - az ábrából leolvashatóan - az idővel egyenesen arányos mennyiség, un. **közepes anomália**, η az **excentrikus anomália**. Képezzük azt a függvényt, aminek a zérushelyét keressük. Ábrázoljuk a kapott függvényt különböző rögzített μ estén.

>
$$f := \eta \rightarrow \eta - e \sin(\eta) - \mu :$$

>
$$e := 0.01674 :$$

>
$$\text{with}(plots) : g := \text{subs}\left(\mu = \frac{\pi}{4}, f(\eta)\right) :$$



A baloldali ábra a $\mu = \frac{\pi}{4}$ értékhez, a jobb oldali ábra különböző μ értékekhez tartozó $f(\eta)$

függvényeket mutatja. A függvénygörbék látszólag 1 meredekségű egyenesek (hiszen e értéke kicsi). A középső, kinagyított ábra cáfolja ezt az állítást. A függvények szigorúan monoton növekvők, teljesülnek a Newton módszer feltételei.

A Kepler-egyenlet megoldásakor a feladat az, hogy adott μ -höz keressük η -t. A Kepler-egyenlet azonban transzcendens, így η -t az μ függvényeként véges formában? megadni nem lehet. Ennek következménye az, hogy a feladatnak az idő függvényeként zárt alakú megoldása nincs. Ez sok nehézséget okoz, különösen a perturbációs számításban.

Kis e esetén az iteráció kezdőértékének $\eta_0 = \mu$ értéknek vehető. Így különböző μ értékekhez kiszámíthatjuk η értékeit, megadott pontossággal.

```
> for j to 5 do  $\mu := evalf\left(\frac{\text{Pi}}{8} + \frac{(j-1) \cdot \pi}{8}\right); \text{Newtoniteral}(f, \mu, 10, 10^{-6})$  end do
 $\mu := 0.392699081699$ 
 $x_1 = 0.399205834120$ 
 $x_2 = 0.399205695664$ 
 $\mu := 0.785398163398$ 
 $x_1 = 0.797376923104$ 
 $x_2 = 0.797376060387$ 
 $\mu := 1.17809724510$ 
 $x_1 = 1.19366270267$ 
 $x_2 = 1.19366081349$ 
 $x_3 = 1.19366081349$ 
 $\mu := 1.57079632680$ 
 $x_1 = 1.58753632680$ 
 $x_2 = 1.58753398201$ 
 $x_3 = 1.58753398201$ 
 $\mu := 1.96349540849$ 
 $x_1 = 1.97886270709$ 
 $x_2 = 1.97886089689$ 
 $x_3 = 1.97886089688$ 
```

Megjegyzés: Az eljárásban az ismeretlent x -szel jelöltük, így $\eta = x$. Az (μ, η) pontokra illeszthető görbékről a 6. fejezetben lesz szó.

A Kepler-egyenletet megoldva ω is kiszámítható az ω és η közti összefüggés alapján. Egyszerű geometria megfontolások alapján (például http://astro.elte.hu/icsip/egi_mechanika/kettest_problema/4fejezet.html)

$$> e := 'e': \tan\left(\frac{\omega}{2}\right) = \sqrt{\frac{1+e}{1-e}} \cdot \tan\left(\frac{\eta}{2}\right) :$$

A Föld esetén

$$> e := 0.01674; \tan\left(\frac{\omega}{2}\right) = \sqrt{\frac{1+e}{1-e}} \cdot \tan\left(\frac{\eta}{2}\right)$$

$$e := 0.01674$$

$$\tan\left(\frac{1}{2} \omega\right) = 1.01688248925 \tan\left(\frac{1}{2} \eta\right)$$

Tehát a két érték közelítőleg megegyezik.


6. Példa Kompresszibilitási együttható R. DeSantis (1976) írta le a valódi gázok kompresszibilitási együtthatóját (z) a következő formában:

$$> z = \frac{1 + x + x^2 - x^3}{(1 - x)^2} :$$

ahol $x = \frac{b}{4 \cdot v}$, b a van der Waals együttható, v a moláris térfogat. Adjuk meg x értékét, ha $z=0.892$.

>
$$f := x \rightarrow z - \frac{1 + x + x^2 - x^3}{(1 - x)^2} : z := 0.892 :$$

Ábrázoljuk a függvényt! Keressük meg a függvény legnagyobb zérushelyét!




Képezzük a függvény deriváltjait, majd válasszunk kezdőpontot!

>
$$\frac{d}{dx} f(x) = D(f)(x); \quad \frac{d^2}{dx^2} f(x) = D^{(2)}(f)(x);$$

$$\text{plot}([D(f)(x), D^{(2)}(f)(x)], x=1.6..1.9, \text{color}=[\text{red}, \text{green}], \text{thickness}=2)$$

$$\frac{d}{dx} \left(0.892 - \frac{1+x+x^2-x^3}{(1-x)^2} \right) = -\frac{1+2x-3x^2}{(1-x)^2} - \frac{2(1+x+x^2-x^3)}{(1-x)^3}$$

$$\frac{d^2}{dx^2} \left(0.892 - \frac{1+x+x^2-x^3}{(1-x)^2} \right) = -\frac{2-6x}{(1-x)^2} - \frac{4(1+2x-3x^2)}{(1-x)^3} - \frac{6(1+x+x^2-x^3)}{(1-x)^4}$$



A második derivált (zöld) negatív, ezért az intervallum bal végpontjából indítsuk az eljárást.

>
$$\text{Newtoniteral}(f, 1.6, 10, 10^{-8})$$

$$x_1 = 1.70572296015$$

$$x_2 = 1.73948435008$$

$$x_3 = 1.74186339420$$

$$x_4 = 1.74187386747$$

$$x_5 = 1.74187386767$$


7. Példa Reynolds szám Lee és Duffy (1976) a következő empirikus képletet találta fibrózusos részecskék szuszpenziója áramlásának sűrűlási tényezőjére:

$$> \frac{1}{\sqrt{x}} = \frac{1}{k} \cdot \ln(RE \cdot \sqrt{x}) + 14 - \frac{5.6}{k} :$$

ahol x a sűrűlási tényező, RE a Reynolds szám, k pedig az oldat koncentrációjától függő állandó. Ha az oldat koncentrációja 0.08 %, $k=0.28$. Mennyi az x értéke, ha $RE=3750$? Képezzük azt a függvényt, amelynek zérushelyét keressük.

$$> f := \text{subs}\left(\{k=0.28, RE=3750\}, \frac{1}{\sqrt{x}} - \frac{1}{k} \cdot \ln(RE \cdot \sqrt{x}) - 14 + \frac{5.6}{k}\right) : f := \text{unapply}(f, x)$$

$$f := x \rightarrow \frac{1}{\sqrt{x}} - 3.57142857143 \ln(3750 \sqrt{x}) + 6.00000000000$$



A függvénygörbe szigorúan monoton csökkenő, konvex görbe, így a feladat megoldásához legyen

$$> x_0 := 0.004 :$$

$$> \text{Newtoniteral}(f, 0.004, 10, 10^{-8})$$

$$x_1 = 0.00494105157810$$

$$x_2 = 0.00511739449449$$

$$x_3 = 0.00512189073926$$

$$x_4 = 0.00512189350260$$


Így 10^{-8} pontossággal megkaptuk a keresett sűrűlási tényezőt.

▼ 2.6. Beépített Maple lehetőségek


A Maple több lehetőséget is kínál a Newton módszer automatikus végrehajtására.

▼ Oktató (Tutorial)

Az Eszközök menüsorban, Oktatók menüpont, Kalkulus-Egy Változó, Newton Eljárás kiválasztásával egyszerűen megkapjuk egy ismeretlenes egyenletek közelítő gyökeit.



Megadhatjuk a függvényt, beállíthatjuk a kezdőpontot, az iterációk számát. Statikus ábrán vagy animációon grafikusán, illetve az Approximate ablakban numerikusan is megjelennek a közelítő értékek.



▼ A Student[Calculus1] csomag NewtonsMethod parancsa

A *Student* csomag több alcsomagot tartalmaz, amely kimondottan egyetemi alapozó matematika témakörök tanulásához és tanításához ad segítséget. A *Calculus I* alcsomag *NewtonMethods* parancsa többféle opciót kínál a közelítő gyök megadására.

> *with(Student[CalculusI])* :

Kötelezően megadandó a függvény (egész pontosan itt kifejezést kell megadni), aminek a zérushelyét keressük, és a kiindulási érték. (Figyelem! A feltételvizsgálatot nem ússzuk meg itt sem.)

Opciók: **output** = *value, sequence, plot*, vagy *animation* (érték, sorozat, ábra, animáció)

> *NewtonMethod*($e^x - 2 + x^2$, $x = 1$, *output* = *value*)

> *NewtonMethod*($e^x - 2 + x^2$, $x = 1$, *output* = *sequence*)

> *NewtonMethod*($e^x - 2 + x^2$, $x = 1$, *output* = *plot*)

> *NewtonMethod*($e^x - 2 + x^2$, $x = 1$, *output* = *animation*)

▼ 2. 7. Feladatok

1. Feladat Adjuk meg az alábbi egyenletek adott intervallumba eső gyökeit 10^{-4} -es hibával!

1. a. $x^3 - 2x^2 - 5 = 0$ $[1, 4]$

1. b. $x - \cos(x) = 0$ $\left[0, \frac{\pi}{2}\right]$

1. c. $x^3 + 3x^2 - 1 = 0$ $[-4, 0]$

2. Feladat Adjuk meg az alábbi egyenletek gyökeit 10^{-5} -es hibával!

2. a. $x = \frac{2 - e^x + x^2}{3}$

2. b. $e^x + 2^{-x} + 2 \cos(x) - 6 = 0$

2. c. $x^2 + 10 \cos(x) = 0$

3. Feladat Határozzuk meg Newton módszere segítségével a következő kifejezések értékeit 4 tizedesjegy pontossággal. (Segítség: írjon fel olyan egyenletet, amelynek megoldása az adott érték.)

3. a. $\sqrt{5}$

3. b. $\ln 2$

3. c. $\sqrt[3]{10}$

4. Feladat Annak valószínűsége, hogy egy teniszjátzmában A játékos 21:0 arányban nyerjen B játékoskal szemben $P = \frac{1+p}{2} \left(\frac{p}{1-p+p^2} \right)^{21}$, ahol p annak a valószínűsége, hogy A megnyer egy játszmát. Adjuk meg 10^{-3} hibával azon p valószínűséget, ami biztosítja azt, hogy A a B -vel játszott játszmák felét 21:0 arányban nyerje.

▼ 2. 8. Ellenőrző kérdések

1. Milyen feltételek teljesülése esetén alkalmazhatjuk a Newton-féle érintőmódszert?

2. Mit biztosít az a feltétel, amely szerint $\frac{d}{dx} f(x) \neq 0$ az $[a, b]$ intervallumon?

3. Miért van szükség a $\frac{d^2}{dx^2} f(x) \neq 0 \ x \in [a, b]$ feltételre?

4. Hogyan kell a kezdőértéket megválasztani?